unittest
{
	assert( [ "2", "0 0", "1 0" ].parse.expand.solve == "0.500000" );
	assert( [ "3", "0 0", "0 1", "1 0" ].parse.expand.solve == "0.707107" );
	assert( [ "10", "10 9", "5 9", "2 0", "0 0", "2 7", "3 3", "2 5", "10 0", "3 7", "1 9" ].parse.expand.solve == "6.726812" );
	
	assert( [ "4", "0 0", "0 1000", "1000 1000", "1000 0" ].parse.expand.solve == "707.106781" );		// 追加ケース
}



import std.algorithm;
import std.conv;
import std.format;
import std.math;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = input.front.to!( long );
	auto xys = new real[ 2 ][ n ];
	foreach( ref xy; xys )
	{
		input.popFront;
		xy = input.front.split.to!( real[] );
	}
	return tuple( xys );
}

auto solve( real[ 2 ][] xys )
{
	// 全ての点を含む円の中心のx座標を半径が一番小さくなるように三分検索
	auto ans = real.max;
	auto l = 0.0L;
	auto r = 1000.0L;
	foreach( i; 0 .. 60 )
	{
		auto lc = l + ( r - l ) / 3;
		auto rc = l + ( r - l ) / 3 * 2;
		
		auto lcd = solve2( xys, lc );
		auto rcd = solve2( xys, rc );
		ans = min( ans, lcd, rcd );
		
		if( lcd < rcd ) r = rc;
		else            l = lc;
	}
	return format( "%0.6f", ans );
}

auto solve2( real[ 2 ][] xys, real x )
{
	// 全ての点を含む円の中心のy座標を半径が一番小さくなるように三分検索
	auto ans = real.max;
	auto l = 0.0L;
	auto r = 1000.0L;
	foreach( i; 0 .. 60 )
	{
		auto lc = l + ( r - l ) / 3;
		auto rc = l + ( r - l ) / 3 * 2;
		
		auto lcd = 0.0L;
		auto rcd = 0.0L;
		foreach( xy; xys )
		{
			lcd = max( lcd, hypot( xy[ 0 ] - x, xy[ 1 ] - lc ) );
			rcd = max( rcd, hypot( xy[ 0 ] - x, xy[ 1 ] - rc ) );
		}
		ans = min( ans, lcd, rcd );
		
		if( lcd < rcd ) r = rc;
		else            l = lc;
	}
	return ans;
}
